Pràctiques Integrades
1er de Matemàtiques

Geometria. Àlgebra lineal.

Curs 2003–04

Índex

III Geometria. Àlgebra lineal. **3**

10 Estudi d'una cònica **3**
10.1 Dibuix de la cònica 3
10.2 Classificació de la cònica 3
10.3 Més exercicis 4

11 Geometria del pla **5**
11.1 Punts i rectes. 5
11.2 Còniques. 7
11.3 Llista d'exercicis. 9
11.4 Altres objectes geomètrics 10

12 Bàsics sobre àlgebra lineal: el paquet LinearAlgebra. **11**
12.1 Introducció de les dades. 11
12.2 Les operacions entre matrius. 12
12.3 La matriu d'un sistema d'equacions lineals. 14
12.4 Les operacions de reducció. 15
12.5 Reduccions automàtiques. 18

Part III
Geometria. Àlgebra lineal.

10 Estudi d'una cònica

En aquesta secció veurem com, a partir d'una cònica determinada per la seva equació, per exemple:

$$x^2 + 2y^2 + \sqrt{8}xy + 4x - \sqrt{8}y + 6 = 0 \tag{1}$$

podem classificar-la, obtenint una cònica equivalent centrada a l'origen i amb un eix l'eix de les x.

10.1 Dibuix de la cònica

En primer lloc observeu que Maple permet dibuixar les solucions d'una equació amb dues variables. Per a això s'utilitza la funció `implicitplot` que està al paquet **plots** (busqueu en l'ajuda del programa quina és la sintaxi adequada d'aquesta comanda).

Exercici 10.1

Dibuixeu la cònica de $\mathbb{R}^2$ de l'equació (1).

10.2 Classificació de la cònica

Per a classificar la cònica, recordeu que primer hem de fer una rotació d'angle α, on α es dedueix de l'equació:

$$\tan(2\alpha) = \frac{\text{coeficient de } xy}{\text{coef. de } y^2 - \text{coef. de } x^2} \tag{2}$$

Exercici 10.2

Escriviu el sinus i el cosinus d'α en funció de $\tan(2\alpha)$. (Primer de tot teniu en compte que $\tan(2\alpha) = \dfrac{\sin(2\alpha)}{\cos(2\alpha)}$ després apliqueu les fòrmules del sinus i el cosinus de l'angle doble i, per acabar, poseu tota la fòrmula en funció del sinus de α).

Per a escriure una rotació d'angle α cal fer la transformació:

$$\begin{array}{rcl} z &=& \cos(\alpha)x - \sin(\alpha)y \\ t &=& \sin(\alpha)x + \cos(\alpha)y \end{array} \quad \text{o bé} \quad \begin{array}{rcl} x &=& \cos(\alpha)z + \sin(\alpha)t \\ y &=& -\sin(\alpha)z + \cos(\alpha)t \end{array}$$

on z i t són el nom dels nous eixos on situem la cònica.

Exercici 10.3

Apliqueu una rotació d'angle α, amb l'α que surt de l'equació (2) a la cònica de l'equació (1) i feu un dibuix de la cònica en funció de les variables z i t.

D'aquesta manera, la cònica actual té l'eix (o els eixos) paral·lel a un dels eixos de coordenades, i no hi ha cap coeficient que multipliqui a zt.

Finalment caldrà traslladar la cònica fins a centrar-la a l'origen (normalment). Per tant, si u i v són les noves coordenades, haurem de fer:

$$\begin{array}{rcl} u & = & z + a \\ v & = & t + b \end{array}$$

on $a = \dfrac{\text{coef. } z}{2 \text{ coef. } z^2}$ i $b = \dfrac{\text{coef. } t}{2 \text{ coef. } t^2}$.

Exercici 10.4

Apliqueu la translació corresponent a la cònica de l'exercici anterior. Podeu classificar la cònica? Doneu els eixos, la directriu i els focus d'aquesta.

Exercici 10.5

Determineu el focus, l'eix i la directriu de la cònica de l'equació (1).

10.3 Més exercicis

Exercici 10.6

Classifiqueu la cònica donada per l'equació:

$$13x^2 + 150x - 171 - 10xy - 102y + 13y^2 = 0$$

Trobeu el/els focus, el/els eixos i el centre.

Exercici 10.7

Classifiqueu la cònica donada per l'equació:

$$3x^2 - 2y^2 - 4xy - 3x + 2y - 4 = 0$$

Trobeu el/els focus, el/els eixos i el centre.

11 Geometria del pla

En aquesta secció analitzarem algunes de les comandes que es troben en el paquet **geometry**. Aquest paquet permet definir i treballar amb objectes geomètrics del pla euclidià. Començarem doncs carregant aquest paquet,

```
> with(geometry):
```

11.1 Punts i rectes.

L'objecte geomètric més senzill que podem definir i utilitzar amb el paquet **geometry** és un punt. Per a definir un punt utilitzem la funció `point(p,x,y)`, on `p` és el nom del punt, `x` és la primera coordenada i `y` la segona.

Exemple 11.1

Si volem definir els punts `A=(1,1)`, `B=(1,2)` i `C=(-1,1)` escriurem les següents comandes.

```
> point(A,1,1);point(B,2,2);point(C,-1,1);
```

El segon objecte que podem definir és una recta, mitjançant la funció `line`. Hi ha dues maneres d'introduir una recta: o bé mitjançant dos punts per on passi, o bé mitjançant una equació.

Exemple 11.2

Podem definir la recta `l1` com la que passa pels punts `A` i `B` que hem definit abans de la següent manera.

```
> line(l1,[A,B]);
```

També podem definir la recta `l2` a partir d'una equació, per exemple $x + y = 12$.

```
> line(l2,x+y=12,[x,y]);
```

El tercer dels arguments de la comanda `line` anterior fa referència al nom que tenen les coordenades dels eixos amb els que treballem. Això no està prefixat i els podem canviar quan volguem. A l'exemple 11.1 anterior automàticament s'ha assignat el nom `x` a l'eix horitzontal i `y` al vertical.

Si volem, podem assignar globalment el nom que volem per a les coordenades redefinint les variables `_EnvHorizontalName` i `_EnvVerticalName`.

Un cop hem definit i posat nom a aquests objectes, podem demanar les propietats que té un d'ells a partir de les funcions:

- `coordinates(p)`; que retorna les coordenades d'un punt `p` com una llista.
- `form(l)`; que retorna la forma de l'objecte geomètric, (i.e., `line2d` si `l` és una recta).
- `Equation(l)`; que retorna l'equació que defineix la recta `l`.

- `detail(l)`; que torna una descripció detallada de la recta `l`.

Exemple 11.3

Redefinim els eixos com `u` i `v` i demanem les propietats de la recta `l1`:

```
> _EnvHorizontalName := u:  _EnvVerticalName := v:
> form(l1);
> Equation(l1);
> detail(l1);
```

Quan tenim definides vàries rectes i punts podem demanar informació de la posició relativa que tenen i fer càlculs mitjançant:

- `AreParallel(l1,l2)`; que retorna `True` si les rectes són paral·leles i `False` si no ho són.
- `ArePerpendicular(l1, l2)`; que retorna `True` si les rectes són perpendiculars i `False` si no ho són.
- `AreConcurrent(l1,l2,l3)`; que retorna `True` si les tres rectes tenen un punt en comú.
- `FindAngle(l1,l2)`; que retorna l'angle que formen dues rectes.
- `intersection(p,l1,l2)`; que guarda a la variable `p` la intersecció entre les dues rectes.
- `ParallelLine(l,p,l1)`; que guarda a la variable `l` la recta que passa per `p` i és paral·lela a `l1`.
- `PerpendicularLine(l,p,l1)`; que guarda a la variable `l` la recta que passa per `p` i és perpendicular a `l1`.
- `projection(q, p, l)`; que guarda a la variable `q` el punt que resulta de fer una projecció perpendicular de `p` a la recta `l`.
- `distance(p,l)`; que calcula la distància entre un punt `p` i una recta `l` o bé entre 2 punts.

Exemple 11.4

```
> AreParallel(l1,l2);
> ArePerpendicular(l1,l2);
> FindAngle(l1,l2);
> intersection(E,l1,l2);
> detail(E);
> intersection(l,l1,l1);
> detail(l);
```

Exercici 11.1

Considereu els punts $A = (2, -3)$, $B = (1, 8)$ i $C = (5, -1)$.

(a) Determineu l'equació de la recta que passa per C i és perpendicular a la recta que passa per A i B.

(b) Quin és el punt d'intersecció d'aquestes dues rectes?

(c) Comproveu si el punt D = $(-4, 3)$ és d'alguna de les dues rectes anteriors.

(d) Calculeu la distància que hi ha entre els punts A, B, C i D.

(e) Quina distància hi ha des del punt C a la recta que passa per A i B?

Finalment, tots els objectes es poden representar gràficament utilitzant la comanda `draw`. Mireu el seu funcionament a l'ajuda de Maple.

Exercici 11.2

Dibuixeu els punts i les rectes de l'exercici anterior.

11.2 Còniques.

A part de punts i rectes, també podem treballar amb objectes més complicats geomètricament com són les còniques. La comanda del paquet **geometry** que permet definir una cònica és `conic()`, que es pot utilitzar de tres formes diferents:

- `conic(p, [A,B,C,E,F], n)`; on la llista `[A,B,C,E,F]` són cinc punts pels quals volem que passi la cònica.
- `conic(p, [dir,foc, exc], n)`; on `dir` ha de ser una objecte `line` que serà la recta directriu de la cònica, `foc` és un objecte `point` que representa un focus de la cònica i `exc` és un valor numèric que representa l'excentricitat.
- `conic(p, eqn, n)`; on `eqn` és una equació quadràtica en dues variables.

En els tres cassos l'argument `p` és el nom amb el que ens referirem a l'objecte que estem definint i l'argument `n` és opcional i és el que determina el nom de les variables respecte les que volem treballar (com en els punts i rectes).
Nota: Una cònica es pot definir com el lloc geomètric dels punts P tals que el quocient de la distància de P a una recta fixada (la directriu de la cònica) amb la distància de P a un punt donat (el focus) és una constant (l'excentricitat). Amb aquest punt de vista les el·lipses són les còniques amb excentricitat menor que 1, les paràboles les d'excentricitat 1 i les hipèrboles les d'excentricitat més gran que 1. Les circumferències corresponen al cas degenerat d'excentricitat 0 amb directriu a *l'infinit.*

Exemple 11.5

```
> eq:=13*x^2 + 150*x- 171 -10*x*y -102*y + 13*y^2 = 0;
> conic(c,eq,[x,y]);
> detail(c);
> draw([c,op(foci(c))(color=blue),center(c)(color=green),
> line(eix,foci(c))(color=yellow),
> PerpendicularLine(eix2,center(c),eix)(color=orange)]);
```

Quan tenim una **el·lipse** podem demanar les següents dades:

- Els dos focus: `foci( )`.
- El centre: `center( )`.
- Les longituds dels eixos major i menor: `MajorAxis( )`, `MinorAxis( )`.
- L'equació: `Equation( )`.

Exemple 11.6

```
> foci(c);
> detail(foci(c));
> simplify(coordinates(foci(c)[1]));
> coordinates(center(c));
> MajorAxis(c);
```

Quan tenim una **hipèrbola** podem demanar les següents dades:

- Els dos focus: `foci( )`.
- El centre: `center( )`.
- Les asímptotes: `asymptotes( )`.
- Els vèrtexs: `vertices( )`.
- L'equació: `Equation( )`.

Exercici 11.3

Comproveu que l'equació $17\,x^2 + 18\,x\,y - 7\,y^2 - 12\,x + 36\,y - 12 = 0$ defineix una hipèrbola i determineu les equacions de les seves asímptotes.

Quan tenim una **paràbola** podem demanar les següents dades:

- El focus: `focus( )`.
- La directriu: `directrix( )`.
- El vèrtex: `vertex( )`.
- L'equació: `Equation( )`.

Exercici 11.4

Comproveu que l'equació $9\,x^2 + 6\,x\,y + y^2 - 4\,x + 12\,y - 4 = 0$ determina una paràbola i calculeu l'equació de la seva directriu.

11.3 Llista d'exercicis.

Exercici 11.5

Determineu el radi d'una circumferència que és tangent a la recta $x + y = 2$ en el punt $(1,\,1)$ i tal que el seu centre està sobre la recta $2\,x - 3\,y = 1$. Feu un dibuix on estiguin representats els elements d'aquest problema.

Exercici 11.6

Definiu la cònica donada per l'equació $x^2 - y^2 = 0$ i dibuixeu-la.

Exercici 11.7

Feu una funció `dibuix` que, a partir d'un objecte cònica `c`, doni com a resultat un dibuix de la cònica amb els seus elements significatius, diferenciant els cassos possibles (focus, centre i eixos per a una el·lipse,...)

Exercici 11.8

Apliqueu la funció `dibuix` que heu disenyat en l'exercici anterior a les còniques següents:

$$9\,x^2+12\,x\,y+4\,y^2-34\,x+38\,y+22=0$$

$$-3\,x^2-8\,x\,y+3\,y^2-28\,x+6\,y+23=0$$

$$\frac{29\,x^2}{3}-2\,x\,y+7\,y^2-\frac{172\,x}{3}-72\,y+\frac{734}{3}=0$$

11.4 Altres objectes geomètrics

Els objectes amb els que pot tractar el paquet **geometry** no estan limitats als punts, rectes o còniques. Si busqueu en l'ajuda de Maple la informació que dóna sobre aquest paquet veureu totes les funcions que té associades.

En particular també és possible treballar amb triangles de forma que es pot resoldre el següent exercici:

Exercici 11.9

Donat el triangle determinat pels vèrtexs $A=(1,3)$, $B=(-2,5)$ i $C=(0,1)$. Determineu i dibuixeu les bisectrius de cada un dels seus angles, les mitjanes i les altures que surten de cada vèrtex i les mediatrius de cada un dels costats. Observeu que cada una d'aquestes famílies de rectes determina un únic punt d'intersecció. Sabeu veure alguna relació d'aliniament entre alguns d'aquests punts?

12 Bàsics sobre àlgebra lineal: el paquet **LinearAlgebra**.

Tot i que Maple pot solucionar sistemes d'equacions lineals utilitzant la comanda `solve( )`, aquesta manera de treballar ens amagarà els càlculs que cal realitzar per a resoldre un sistema d'equacions lineals i en alguns casos pot portar-nos a solucions incorrectes.

Per exemple, si volem resoldre el sistema d'equacions homogeni depenent del paràmetre a donat per les equacions

$$\left\{ \begin{array}{r} a\,x + (a-1)\,y = 0 \\ x - a\,y = 0 \end{array} \right.$$

utilitzant la comanda `solve` farem:

```
> equacions:={a*x+(a-1)*y=0,x-a*y=0};
> solve(equacions,{x,y});
```

però si considerem el cas en el que a val $\dfrac{-1+\sqrt{5}}{2}$ resulta que tenim

```
> caspart:= subs(a=(-1+sqrt(5))/2,equacions);
```

Que té una infinitat de solucions, com es pot veure amb:

```
> solve(caspart,y);
```

I encara que considerem el paràmetre a com una altra incògnita el resultat que ens dóna `solve( )` no és del tot satisfactori.

```
> solve(equacions);
```

Per tant és molt important conéixer quins són aquests càlculs i saber com funcionen per a tractar aquest tipus de problemes. També cal tenir en compte que per a entendre conceptes més profunds com són els relacionats amb els espais vectorials cal saber solucionar sistemes d'equacions lineals de tota mena. Els diferents tipus d'operacions amb matrius són la maquinaria bàsica per a resoldre sistemes d'equacions lineals. Per a poder realitzar tots aquests tipus de càlculs és possible utilitzar les funcions que defineix el paquet de Maple **LinearAlgebra**. Aquest paquet de Maple no és l'únic que existeix per a realitzar aquest tipus de càlculs (per exemple, en la instal·lació normal de Maple també trobareu el paquet **linalg**) i les mateixes operacions es poden realitzar amb instruccions de nom i sintaxi diferents per a cada paquet. A continuació presentem les que són més fonamentals per a treballar amb problemes típics d'àlgebra lineal utilitzant **LinearAlgebra**.

Com sempre, comencem *carregant* el paquet:

```
> restart;
> with(LinearAlgebra):
```

12.1 Introducció de les dades.

Els objectes sobre els que treballen les funcions del paquet `LinearAlgebra` són les matrius i els vectors. Per a introduir una matriu o un vector podem utilitzar la notació `< , , , ...>` o `< | | | ...>` segons si volem fer referència a les files de l'objecte (notació amb comes) o a les columnes (notació amb barres).

Exemple 12.1

La matriu $M = \begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{pmatrix}$ es pot introduir per files com

```
> Mf:=< < 1 | 2 | 3 >,< 4 | 5 |6 > >;
```

o per columnes com

```
> Mc:= < < 1 , 4> | < 2 , 5 > | < 3 , 6 > >;
```

Fixeu-vos que en els dos casos es veu el mateix resultat exactament.

Aquestes construccions només són formes abreujades i simplificades d'introduir les matrius o vectors, podeu consultar a l'ajuda del programa les funcions per a escriure les matrius i els vectors que són `Matrix( )` i `Vector( )` i podreu veure totes les possibilitats que tenen (i també que és una mica més llarg tenir definida una matriu a partir d'aquestes comandes).

Per a extreure una submatriu d'una matriu ja definida podem utilitzar la sintaxi usual a les llistes, per exemple, si `A` és una matriu, `A[i,j]` és el coeficient de la fila `i`, columna `j`. Tenint en compte que tant l'argument `i` com el `j` poden ser rangs i llistes.

Exemple 12.2

Considereu la matriu `Mf` de l'exemple anterior i proveu:

```
> Mf[2,1];
> Mf[1..2,2..3];
> Mf[[2,1],1..3];
```

També heu de saber que el paquet `LinearAlgebra` té funcions per a introduir molts tipus especials de matrius amb poc esforç (matrius diagonals, triangulars superiors, amb tots els coeficients iguals, ...). Pot ser ara és un bon moment per a fer un cop d'ull al joc de funcions d'aquest paquet mirant la secció de l'ajuda del programa dedicada al paquet **LinearAlgebra** on hi trobareu la llista completa.

12.2 Les operacions entre matrius.

Les operacions bàsiques entre matrius són la suma (i la diferència), el producte per un escalar i el producte de matrius. Quan utilitzem el paquet `LinearAlgebra` totes aquestes operacions estan disponibles de forma immediata. Podeu comprovar-ho en els exemples següents.

Exemple 12.3

```
> M1:= < <1,2,-1>|<0,3,3>|<-1,1,-3> >;
> M2:= < <0,1,-1>|<-2,3,5>|<1,1,0> >;
> M1+M2;
> M1-M2;
> 3*M1;
> 2*M1-4*M2;
```

Només cal tenir en compte que, com que el producte de matrius no es fa multiplicant cada una de les components dels factors, l'operador `*` (asterisc) no produeix el resultat que esperaríem en un principi (de fet dóna un missatge d'error). Per a obtenir el vertader producte de matrius cal utilitzar l'operador `.` (punt).

Exemple 12.4

```
> M1.M2;
> M1*M2;
> M2.M1;
```

El producte de matrius no sempre es pot efectuar. Si les matrius que intenteu multiplicar no són dels tamanys adequats, Maple us donarà un missatge d'error.

Exemple 12.5

```
> M3:= < <1,2>|<2,-1>|<1,-1> >;
> M1.M3;
```

En canvi sí que es pot realitzar l'operació amb l'ordre invers.

```
> M3.M1;
```

Per a practicar una mica més amb la introducció de dades en vectors i matrius i en la realització dels càlculs elementals amb matrius aquí teniu un petit exercici.

Exercici 12.1

Introduïu la matriu $P_1 = \begin{pmatrix} \frac{4}{7} & \frac{1}{7} & -\frac{1}{7} \\ -\frac{5}{21} & \frac{4}{21} & \frac{1}{7} \\ -\frac{3}{7} & \frac{1}{7} & -\frac{1}{7} \end{pmatrix}$, el vector $v_1 = \begin{pmatrix} 1 \\ -\frac{11}{3} \\ -2 \end{pmatrix}$ i el vector $v_2 = \begin{pmatrix} 1 \\ -1 \\ 2+2\sqrt{2} \end{pmatrix}$.

Utilitzant la matriu $M_1 = \begin{pmatrix} 1 & 0 & -1 \\ 2 & 3 & 1 \\ -1 & 3 & -3 \end{pmatrix}$, que ja hauríeu de tenir introduïda en l'exemple 12.3 d'aquest mateix apartat, calculeu els productes $(M_1)\ .\ (P_1)$, $(M_1)\ .\ (v_i)$, $(P_1)\ .\ (v_i)$ (per a $i = 1$ i 2). Quina relació creieu que hi ha entre els productes $(M_1)\ .\ (v_i)$, $(P_1)\ .\ (v_i)$ i els vectors v_i? Qué passa si intenteu realitzar el producte $(v_1)\ .\ (M_1)$? Busqueu en l'ajuda de Maple com es pot transposar un vector o una matriu i calculeu el producte del vector fila amb les mateixes components que v_1 per la matriu M_1.

12.3 La matriu d'un sistema d'equacions lineals.

Quan es vol discutir i resoldre un sistema d'equacions lineals, el procediment usual consisteix en obtenir un sistema equivalent (és a dir, amb el mateix conjunt de solucions) que tingui una forma *reduïda.* (El significat de reduït depèn del context. En general podem dir que un sistema és reduït si podem dir sense cap mena de dubte si té solucions o no i en el cas que en tingui puguem llegir, sense fer càlculs addicionals, quines són aquestes solucions). Aquest procediment de *reducció* sempre es podrà realitzar utilitzant tres tipus bàsics de transformacions:

- Intercanviar dues de les equacions entre si.
- Multiplicar una de les equacions per una constant no nul·la.
- Sumar a una de les equacions un múltiple d'una de les altres.

Recordeu que per a poder fer aquestes operacions només necessitem treballar amb els *coeficients* de les equacions, sense utilitzar per res les *incògnites.* Així, per exemple, si es vol treballar sobre el sistema d'equacions $\left\{\begin{array}{c} 3\,x-2\,y=2 \\ 2\,x-y=7 \end{array}\right.$ bastarà que treballem sobre la matriu $\begin{pmatrix} 3 & -2 & 2 \\ 2 & -1 & 7 \end{pmatrix}$.

Maple ens permet fer aquest procés de forma automàtica. Per això el paquet **LinearAlgebra** proporciona la comanda `GenerateMatrix( )` (i la comanda inversa `GenerateEquations( )`).

Exemple 12.6

Per a obtenir la matriu de coeficients i el terme independent del sistema d'equacions

$$\left\{\begin{array}{c} x-y+z-2\,t=1 \\ y-2\,z=6 \\ 2\,x+4\,y+t=8 \\ 44\,x-y+34\,z-21\,t=98 \end{array}\right.$$

posarem

```
> sistema1:=[x-y+z-2*t=1, y-2*z=6, 2*x+4*y+t=8,44*x-y+34*z-21*t=98];
> vars:=[x,y,z,t];
> S1:= GenerateMatrix(sistema1,vars);
```

Noteu que `S1` és una llista de dues matrius, la primera amb els coeficients i l'altra amb el terme independent. Si el que voleu és la matriu ampliada del sistema podeu fer

```
> Sa1:= <S1[1]|S1[2]>;
```

o bé utilitzar l'opció de la comanda `GenerateMatrix( )` que fa el procés anterior d'un sol cop

```
> GenerateMatrix(sistema1,vars,augmented);
```

Noteu que és molt important especificar quines són les incògnites ja que, si no es fa, Maple no seria capaç de distingir en un sistema d'equacions lineals que depèn d'un paràmetre si les indeterminades que hi surten són totes incògnites o no.

Exemple 12.7

Si tenim el sistema d'equacions, depenent del paràmetre λ, donat per $\left\{ \begin{array}{r} x - 2\,y = \lambda \\ 2\,x - 3\,y = 2 - \lambda \end{array} \right.$, per a obtenir la seva matriu de coeficients haurem de fer

```
> GenerateMatrix([x-2*y=lambda,2*x-3*y=2-lambda],[x,y]);
```

encara que dins d'aquest sistema la dependència respecte λ també sigui lineal.

Si no posem quines són les incògnites Maple donarà un missatge d'error.

```
> GenerateMatrix([x-2*y=lambda,2*x-3*y=2-lambda]);
```

L'exemple següent mostra el procediment per a generar un sistema d'equacions a partir de la matriu de coeficients utilitzant `GenerateEquations( )`. Els detalls sobre la comanda `GenerateEquations( )` els podreu trobar a l'ajuda de Maple en l'apartat corresponent a la funció `GenerateEquations` del paquet **LinearAlgebra**.

Exemple 12.8

El sistema d'equacions lineals que té per matriu de coeficients $\begin{pmatrix} 1 & -2 & 3 \\ 2 & 3 & 1 \\ -1 & 1 & -5 \end{pmatrix}$ i per terme independent $\begin{pmatrix} 2 \\ -1 \\ 7 \end{pmatrix}$ respecte les incògnites $\begin{pmatrix} x \\ y \\ z \end{pmatrix}$ s'obtindrà amb la comanda:

```
> eqs:=GenerateEquations(<<1,2,-1>|<-2,3,1>|<3,1,-5>>,[x,y,z],<2,-1,7>);
```

El resultat serà una llista amb les tres equacions que volíem, de la que podem extreure cada un dels seus elements amb la notació usual `eqs[i]` per a `i=1..3`.

12.4 Les operacions de reducció.

La comanda de Maple `RowOperation( )` permet realitzar **totes** les operacions de reducció de la matriu d'un sistema d'equacions lineals. En els exemples que venen tot seguit podeu veure una mostra de cada un dels tipus aplicat sobre la matriu $\begin{pmatrix} 2 & 1 & -1 \\ 3 & 5 & -2 \\ 4 & 8 & 16 \end{pmatrix}$.

```
> matriu:= < <2|1|-1>,<3|5|-2>,<4|8|16> >;
```

Exemple 12.9

Primer intercanviem la primera fila amb la tercera:

```
> RowOperation(matriu,[1,3]);
```

Fixeu-vos que els arguments són: la matriu sobre la que volem actuar i una llista amb el parell files que voleu intercanviar.

Exemple 12.10

També podem multiplicar tota la tercera fila per $\dfrac{1}{4}$:

```
> RowOperation(matriu,3,1/4);
```

Aquí heu de notar que els arguments són: la matriu sobre la que volem actuar, la fila sobre la que volem fer la operació i finalment el nombre pel que volem multiplicar.

Exemple 12.11

I finalment podem sumar a la segona fila la primera multiplicada per $\dfrac{-3}{2}$ (cosa que *elimina* la primera incògnita en la segona equació).

```
> RowOperation(matriu,[2,1],-3/2);
```

En aquest cas els arguments són: la matriu sobre la que actuem, una llista amb dos nombres que designen, el primer, la fila sobre la que volem realitzar l'operació i, el segon, la fila que utilitzem per actuar sobre la primera i, com a últim argument, el nombre pel que volem multiplicar.

Cal dir que en tots aquests casos també es podria afegir un argument opcional (`inplace`) que té l'efecte de realitzar les operacions i modificar la mateixa matriu sobre la que les fem d'acord amb el resultat obtingut (fixeu-vos que utilitzar aquesta opció té el perill d'equivocar-se amb l'operació que es vol realitzar i no saber, o poder, desfer els efectes no desitjats).

Exemple 12.12

Podem modificar `matriu` per a deixar una matriu reduïda realitzant les operacions següents:

```
> RowOperation(matriu,[2,1],-3/2,inplace);
> RowOperation(matriu,[3,1],-2,inplace);
> RowOperation(matriu,[3,2],-12/7,inplace);
> RowOperation(matriu,1,1/2,inplace);
> RowOperation(matriu,2,2/7,inplace);
> RowOperation(matriu,3,7/132,inplace);
```

També es pot dir que existeix la comanda `ColumnOperation( )` amb les mateixes característiques que la comanda de les operacions per files però aplicant les operacions a les columnes.

Ara ja podeu mirar de fer els següents exercicis:

Exercici 12.2

Donat el sistema d'equacions $\left\{\begin{array}{c} 2\,x_1 - 4\,x_2 + 3\,x_3 - 5\,x_4 + x_5 = 3 \\ x_1 - 4\,x_2 - 2\,x_3 - 7\,x_5 = 2 \\ 5\,x_1 - 3\,x_3 - 2\,x_4 + 2\,x_5 = 1 \\ x_2 - 4\,x_3 + 3\,x_4 - x_5 = \dfrac{3}{2} \end{array}\right.$ genereu la seva matriu ampliada, realitzeu les operacions de reducció necessàries en aquesta matriu i determineu si té solucions o no en té. Si en té, com són aquestes solucions?

Exercici 12.3

En la fabricació d'un cert producte alimentari s'utilitzen quatre components diferents. Una unitat de cada un dels ingredients conté les quantitats de vitamines (miligrams) i de kilocalories que s'expressen en la taula següent:

	Comp1	Comp2	Comp3	Comp4
vit A	1	1	1	2
vit B	1	2	1	3
vit C	1	3	2	1
Kilocal	2	2	1	1

Quines quantitats de cada un dels ingredients s'han de barrejar per a produir un producte final que aporti, exactament, 300 mg de vitamina A, 430 mg de vitamina B, 310 mg de vitamina C i tingui un valor energètic de 250 kilocalories?

És possible fabricar un producte final en el que la combinació de vitamines i d'aport energètic sigui qualsevol?

Exercici 12.4

Estudieu, per als diferents valors dels paràmetres b i t, el sistema d'equacions lineals respecte $[x,\, y,\, z]$.

$$\begin{cases} 3\,x + 2\,y + z = t \\ x - y + 2\,z = 1 + t^2 \\ 3\,x + 7\,y - 4\,z = -1 - t - t^2 - t^3 \\ 2\,x + y + b\,z = t^3 \end{cases}$$

Sovint no serà necessari realitzar les operacions de reducció sobre la matriu d'un sistema d'equacions lineals una per una. La comanda de Maple `Pivot( )` és en aquests casos força útil. Quan executem la comanda `Pivot(A,i,j)`, se sumen a totes les files de la matriu A múltiples adequats de la fila que ocupa la posició i per tal que els coeficients en la columna j siguin tots nuls (naturalment la fila i no es modifica, tampoc es pot realitzar la operació si el coeficient en la columna j de la fila i és 0). Aquesta comanda també té l'opció `inplace` per a realitzar els càlculs modificant la matriu sobre la que estem treballant.

Exemple 12.13

Si es considera la matriu donada per $A = \begin{pmatrix} 1 & 3 & 0 \\ 2 & -1 & -5 \\ 0 & 1 & 3 \end{pmatrix}$

```
> A:= <<1|3|0>,<2|-1|-5>,<0|1|3>>;
```

Podem *pivotar* sobre l'element [3, 2] amb

```
> Pivot(A,3,2);
```

Però no ho podrem fer sobre l'element [3, 1].

```
> Pivot(A,3,1);
```

12.5 Reduccions automàtiques.

Quan no hi ha cap indeterminació en el camí per a fer la reducció d'un sistema d'equacions lineals es pot demanar a Maple que redueixi d'un sol cop la matriu del sistema.

La comanda de Maple `GaussianElimination( )` és la que realitza la reducció *Gaussiana* (dóna com a resultat una matriu reduida amb zeros a la part inferior)

Exemple 12.14

Apliquem `GaussianElimination( )` sobre la matriu $A = \begin{pmatrix} 1 & 3 & 0 \\ 2 & -1 & -5 \\ 0 & 1 & 3 \end{pmatrix}$ (que és la de l'exemple anterior).

```
> GaussianElimination(A);
```

Per a reduir completament la matriu del sistema d'equacions lineals tenim `ReducedRowEchelonForm( )` que a part de la reducció Gaussiana torna com a resultat una matriu en la que els *pivots* són 1 i tots els altres coeficients de la columna d'un d'aquests pivots són 0.

Exemple 12.15

```
> ReducedRowEchelonForm(A);
```

Finalment, si el que es vol és determinar la solució d'un sistema d'equacions lineals donat per la seva matriu i pel terme independent, es pot utilitzar la comanda `LinearSolve( )`.

Exemple 12.16

La solució del sistema d'equacions amb matriu de coeficients A i terme independent $\begin{pmatrix} 1 \\ 2 \\ -5 \end{pmatrix}$ es pot obtenir amb

```
> LinearSolve(A,<1,2,-5>);
```

Encara que el sistema sigui indeterminat, la comanda `LinearSolve( )` també funciona

```
> LinearSolve(<<1|2|1>,<-2|1|-5>>,<2,-1>);
```